

R1. <Rat22F> ::= <Opt Function Definitions> \$ <Opt Declaration List> <Statement List> \$

R2. <Opt Function Definitions> ::= <Function Definitions> | <Empty>

R3.

<Function Definitions> ::= <Function> <Function Definitions Prime>

<Function Definitions Prime> ::= <Function Definitions> | <Empty>

R4. <Function> ::= function <Identifier> (<Opt Parameter List>) <Opt Declaration List> <Body>

R5. <Opt Parameter List> ::= <Parameter List> | <Empty>

R6.

<Parameter List> ::= <Parameter> <Parameter List Prime>

<Parameter List Prime> ::= , <Parameter List> | <Empty>

R7. <Parameter> ::= <IDs> <Qualifier>

R8. <Qualifier> ::= integer | boolean | real

R9. <Body> ::= { <Statement List> }

R10. <Opt Declaration List> ::= <Declaration List> | <Empty>

R11.

<Declaration List> ::= <Declaration> ; <Declaration List Prime>

<Declaration List Prime> ::= <Declaration List> | <Empty>

R12. <Declaration> ::= <Qualifier> <IDs>

R13.

<IDs> ::= <Identifier> <IDs Prime>

<IDs Prime> ::= , <IDs> | <Empty>

R14.

<Statement List> ::= <Statement> <Statement List Prime>

<Statement List Prime> ::= <Statement List> | <Empty>

R15. <Statement> ::= <Compound> | <Assign> | <If> | <Return> | <Print> | <Scan> | <While>

R16. <Compound> ::= { <Statement List> }

R17. <Assign> ::= <Identifier> = <Expression> ;

R18.

<If> ::= if (<Condition>) <Statement> <If Prime>

<If Prime> ::= endif | else <Statement> endif

R19.

<Return> ::= return <Return Prime>

<Return Prime> ::= ; | <Expression> ;

R21. <Print> ::= put (<Expression>);

R21. <Scan> ::= get (<IDs>);

R22. <While> ::= while (<Condition>) <Statement>

R23. <Condition> ::= <Expression> <Relop> <Expression>

R24. <Relop> ::= == | != | > | < | <= | >=

R25.

<Expression> ::= <Term> <Expression Prime>

<Expression Prime> ::= + <Term> <Expression Prime> | - <Term> <Expression Prime> | <Empty>

R26.

<Term> ::= <Factor> <Term Prime>

<Term Prime> ::= * <Factor> <Term Prime> | / <Factor> <Term Prime> | <Empty>

R27. <Factor> ::= - <Primary> | <Primary>

R28.

<Primary> ::= <Identifier> <Primary Prime> | <Integer> | (<Expression>) | <Real> | true | false

<Primary Prime> ::= [<IDs>] | <Empty>

R29. <Empty> ::=

Note: <Identifier>, <Integer>, <Real> are token types as defined in section (1) above